

NAG Library Routine Document

G13FGF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G13FGF estimates the parameters of a univariate regression-exponential GARCH(p, q) process (see Engle and Ng (1993)).

2 Specification

```
SUBROUTINE G13FGF (DIST, YT, X, LDX, NUM, IP, IQ, NREG, MN, NPAR, THETA,      &
                   SE, SC, COVR, LDCOVR, HP, ET, HT, LGF, COPTS, MAXIT,      &
                   TOL, WORK, LWORK, IFAIL)
INTEGER          LDX, NUM, IP, IQ, NREG, MN, NPAR, LDCOVR, MAXIT,      &
                  LWORK, IFAIL
REAL (KIND=nag_wp) YT(NUM), X(LDX,*), THETA(NPAR), SE(NPAR), SC(NPAR),      &
                  COVR(LDCOVR,NPAR), HP, ET(NUM), HT(NUM), LGF, TOL,      &
                  WORK(LWORK)
LOGICAL          COPTS
CHARACTER(1)    DIST
```

3 Description

A univariate regression-exponential GARCH(p, q) process, with q coefficients α_i , for $i = 1, 2, \dots, q$, q coefficients ϕ_i , for $i = 1, 2, \dots, q$, p coefficients, β_i , for $i = 1, 2, \dots, p$, and k linear regression coefficients b_i , for $i = 1, 2, \dots, k$, can be represented by:

$$\begin{aligned} y_t &= b_o + x_t^T b + \epsilon_t \\ \ln(h_t) &= \alpha_0 + \sum_{i=1}^q \alpha_i z_{t-i} + \sum_{i=1}^q \phi_i (|z_{t-i}| - E[|z_{t-i}|]) + \sum_{i=1}^p \beta_i \ln(h_{t-i}), \quad t = 1, 2, \dots, T \end{aligned} \quad (1)$$

where $z_t = \frac{\epsilon_t}{\sqrt{h_t}}$, $E[|z_{t-i}|]$ denotes the expected value of $|z_{t-i}|$ and $\epsilon_t | \psi_{t-1} = N(0, h_t)$ or $\epsilon_t | \psi_{t-1} = S_t(df, h_t)$. Here S_t is a standardized Student's t -distribution with df degrees of freedom and variance h_t , T is the number of terms in the sequence, y_t denotes the endogenous variables, x_t the exogenous variables, b_o the regression mean, b the regression coefficients, ϵ_t the residuals, h_t the conditional variance, df the number of degrees of freedom of the Student's t -distribution, and ψ_t the set of all information up to time t .

G13FGF provides an estimate $\hat{\theta}$, for the vector $\theta = (b_o, b^T, \omega^T)$ where $b^T = (b_1, \dots, b_k)$, $\omega^T = (\alpha_0, \alpha_1, \dots, \alpha_q, \phi_1, \dots, \phi_q, \beta_1, \dots, \beta_p, \gamma)$ when $DIST = 'N'$, and $\omega^T = (\alpha_0, \alpha_1, \dots, \alpha_q, \phi_1, \dots, \phi_q, \beta_1, \dots, \beta_p, \gamma, df)$ when $DIST = 'T'$.

MN, NREG can be used to simplify the GARCH(p, q) expression in (1) as follows:

No Regression and No Mean

$$y_t = \epsilon_t,$$

$$MN = 0,$$

$$NREG = 0 \text{ and}$$

θ is a $(2 \times q + p + 1)$ vector when $DIST = 'N'$, and a $(2 \times q + p + 2)$ vector, when $DIST = 'T'$.

No Regression

$$y_t = b_o + \epsilon_t,$$

$$\text{MN} = 1,$$

$$\text{NREG} = 0 \text{ and}$$

θ is a $(2 \times q + p + 2)$ vector when DIST = 'N' and a $(2 \times q + p + 3)$ vector, when DIST = 'T'.

Note: if the $y_t = \mu + \epsilon_t$, where μ is known (not to be estimated by G13FGF) then (1) can be written as $y_t^\mu = \epsilon_t$, where $y_t^\mu = y_t - \mu$. This corresponds to the case **No Regression and No Mean**, with y_t replaced by $y_t - \mu$.

No Mean

$$y_t = x_t^T b + \epsilon_t,$$

$$\text{MN} = 0,$$

$$\text{NREG} = k \text{ and}$$

θ is a $(2 \times q + p + 1 + k)$ vector when DIST = 'N' and a $(2 \times q + p + 2 + k)$ vector, when DIST = 'T'.

4 References

Bollerslev T (1986) Generalised autoregressive conditional heteroskedasticity *Journal of Econometrics* **31** 307–327

Engle R (1982) Autoregressive conditional heteroskedasticity with estimates of the variance of United Kingdom inflation *Econometrica* **50** 987–1008

Engle R and Ng V (1993) Measuring and testing the impact of news on volatility *Journal of Finance* **48** 1749–1777

Glosten L, Jagannathan R and Runkle D (1993) Relationship between the expected value and the volatility of nominal excess return on stocks *Journal of Finance* **48** 1779–1801

Hamilton J (1994) *Time Series Analysis* Princeton University Press

5 Parameters

1: DIST – CHARACTER(1) *Input*

On entry: the type of distribution to use for e_t .

DIST = 'N'

A Normal distribution is used.

DIST = 'T'

A Student's t -distribution is used.

Constraint: DIST = 'N' or 'T'.

2: YT(NUM) – REAL (KIND=nag_wp) array *Input*

On entry: the sequence of observations, y_t , for $t = 1, 2, \dots, T$.

3: X(LDX,*) – REAL (KIND=nag_wp) array *Input*

Note: the second dimension of the array X must be at least NREG.

On entry: row t of X must contain the time dependent exogenous vector x_t , where $x_t^T = (x_t^1, \dots, x_t^k)$, for $t = 1, 2, \dots, T$.

4:	LDX – INTEGER	<i>Input</i>
<i>On entry:</i> the first dimension of the array X as declared in the (sub)program from which G13FGF is called.		
<i>Constraint:</i> $LDX \geq NUM$.		
5:	NUM – INTEGER	<i>Input</i>
<i>On entry:</i> T , the number of terms in the sequence.		
<i>Constraints:</i>		
$NUM \geq \max(IP, IQ);$		
$NUM \geq NREG + MN$.		
6:	IP – INTEGER	<i>Input</i>
<i>On entry:</i> the number of coefficients, β_i , for $i = 1, 2, \dots, p$.		
<i>Constraint:</i> $IP \geq 0$ (see also NPAR).		
7:	IQ – INTEGER	<i>Input</i>
<i>On entry:</i> the number of coefficients, α_i , for $i = 1, 2, \dots, q$.		
<i>Constraint:</i> $IQ \geq 1$ (see also NPAR).		
8:	NREG – INTEGER	<i>Input</i>
<i>On entry:</i> k , the number of regression coefficients.		
<i>Constraint:</i> $NREG \geq 0$ (see also NPAR).		
9:	MN – INTEGER	<i>Input</i>
<i>On entry:</i> if $MN = 1$, the mean term b_0 will be included in the model.		
<i>Constraint:</i> $MN = 0$ or 1 .		
10:	NPAR – INTEGER	<i>Input</i>
<i>On entry:</i> the number of parameters to be included in the model. $NPAR = 1 + 2 \times IQ + IP + MN + NREG$ when $DIST = 'N'$ and $NPAR = 2 + 2 \times IQ + IP + MN + NREG$ when $DIST = 'T'$.		
<i>Constraint:</i> $NPAR < 20$.		
11:	THETA(NPAR) – REAL (KIND=nag_wp) array	<i>Input/Output</i>
<i>On entry:</i> the initial parameter estimates for the vector θ .		
The first element must contain the coefficient α_o and the next IQ elements must contain the autoregressive coefficients α_i , for $i = 1, 2, \dots, q$.		
The next IQ elements contain the coefficients ϕ_i , for $i = 1, 2, \dots, q$.		
The next IP elements must contain the moving average coefficients β_i , for $i = 1, 2, \dots, p$.		
If $DIST = 'T'$, the next element must contain an estimate for df , the number of degrees of freedom of the Student's t -distribution.		
If $MN = 1$, the next element must contain the mean term b_o .		
If COPTS = .FALSE., the remaining $NREG$ elements are taken as initial estimates of the linear regression coefficients b_i , for $i = 1, 2, \dots, k$.		
<i>On exit:</i> the estimated values $\hat{\theta}$ for the vector θ .		

The first element contains the coefficient α_o and the next IQ elements contain the coefficients α_i , for $i = 1, 2, \dots, q$.

The next IQ elements contain the coefficients ϕ_i , for $i = 1, 2, \dots, q$.

The next IP elements are the moving average coefficients β_i , for $i = 1, 2, \dots, p$.

If DIST = 'T', the next element contains an estimate for df then the number of degrees of freedom of the Student's t -distribution.

If MN = 1, the next element contains an estimate for the mean term b_o .

The final NREG elements are the estimated linear regression coefficients b_i , for $i = 1, 2, \dots, k$.

12: SE(NPAR) – REAL (KIND=nag_wp) array Output

On exit: the standard errors for $\hat{\theta}$.

The first element contains the standard error for α_o and the next IQ elements contain the standard errors for α_i , for $i = 1, 2, \dots, q$. The next IQ elements contain the standard errors for ϕ_i , for $i = 1, 2, \dots, q$. The next IP elements are the standard errors for β_j , for $j = 1, 2, \dots, p$.

If DIST = 'T', the next element contains the standard error for df , the number of degrees of freedom of the Student's t -distribution.

If MN = 1, the next element contains the standard error for b_o .

The final NREG elements are the standard errors for b_j , for $j = 1, 2, \dots, k$.

13: SC(NPAR) – REAL (KIND=nag_wp) array Output

On exit: the scores for $\hat{\theta}$.

The first element contains the scores for α_o , the next IQ elements contain the scores for α_i , for $i = 1, 2, \dots, q$, the next IQ elements contain the scores for ϕ_i , for $i = 1, 2, \dots, q$, the next IP elements are the scores for β_j , for $j = 1, 2, \dots, p$.

If DIST = 'T', the next element contains the scores for df , the number of degrees of freedom of the Student's t -distribution.

If MN = 1, the next element contains the score for b_o .

The final NREG elements are the scores for b_j , for $j = 1, 2, \dots, k$.

14: COVR(LDCOVR,NPAR) – REAL (KIND=nag_wp) array Output

On exit: the covariance matrix of the parameter estimates $\hat{\theta}$, that is the inverse of the Fisher Information Matrix.

15: LDCOVR – INTEGER Input

On entry: the first dimension of the array COVR as declared in the (sub)program from which G13FGF is called.

Constraint: $LDCOVR \geq NPAR$.

16: HP – REAL (KIND=nag_wp) Input/Output

On entry: if COPTS = .FALSE. then HP is the value to be used for the pre-observed conditional variance, otherwise HP is not referenced.

On exit: if COPTS = .TRUE. then HP is the estimated value of the pre-observed conditional variance.

17: ET(NUM) – REAL (KIND=nag_wp) array Output

On exit: the estimated residuals, ϵ_t , for $t = 1, 2, \dots, T$.

18:	HT(NUM) – REAL (KIND=nag_wp) array	<i>Output</i>
	<i>On exit:</i> the estimated conditional variances, h_t , for $t = 1, 2, \dots, T$.	
19:	LGF – REAL (KIND=nag_wp)	<i>Output</i>
	<i>On exit:</i> the value of the log-likelihood function at $\hat{\theta}$.	
20:	COPTS – LOGICAL	<i>Input</i>
	<i>On entry:</i> if COPTS = .TRUE., the routine provides initial parameter estimates of the regression terms, otherwise these are provided by you.	
21:	MAXIT – INTEGER	<i>Input</i>
	<i>On entry:</i> the maximum number of iterations to be used by the optimization routine when estimating the GARCH(p, q) parameters.	
	<i>Constraint:</i> MAXIT > 0.	
22:	TOL – REAL (KIND=nag_wp)	<i>Input</i>
	<i>On entry:</i> the tolerance to be used by the optimization routine when estimating the GARCH(p, q) parameters.	
23:	WORK(LWORK) – REAL (KIND=nag_wp) array	<i>Workspace</i>
24:	LWORK – INTEGER	<i>Input</i>
	<i>On entry:</i> the dimension of the array WORK as declared in the (sub)program from which G13FGF is called.	
	<i>Constraint:</i> LWORK $\geq (NREG + 3) \times NUM + 3$.	
25:	IFAIL – INTEGER	<i>Input/Output</i>
	<i>On entry:</i> IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.	
	For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL $\neq 0$ on exit, the recommended value is -1. When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.	
	<i>On exit:</i> IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).	

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Note: G13FGF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

IFAIL = 1

On entry, NREG < 0,
or MN > 1,
or MN < 0,
or IQ < 1,
or IP < 0,
or NPAR ≥ 20 ,

or NPAR has an invalid value,
 or LDCOVR < NPAR,
 or LDX < NUM,
 or DIST ≠ 'N',
 or DIST ≠ 'T',
 or MAXIT ≤ 0,
 or NUM < max(IP, IQ),
 or NUM < NREG + MN.

IFAIL = 2

On entry, LWORK < (NREG + 3) × NUM + 3.

IFAIL = 3

The matrix X is not full rank.

IFAIL = 4

The information matrix is not positive definite.

IFAIL = 5

The maximum number of iterations has been reached.

IFAIL = 6

The log-likelihood cannot be optimized any further.

IFAIL = 7

No feasible model parameters could be found.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.8 in the Essential Introduction for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.7 in the Essential Introduction for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.6 in the Essential Introduction for further information.

7 Accuracy

Not applicable.

8 Parallelism and Performance

G13FGF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

G13FGF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

None.

10 Example

This example fits a GARCH(1,2) model with Student's t -distributed residuals to some simulated data.

The process parameter estimates, $\hat{\theta}$, are obtained using G13FGF, and a four step ahead volatility estimate is computed using G13FHF.

The data was simulated using G05PGF.

10.1 Program Text

```
Program g13fgfe

!     G13FGF Example Program Text

!     Mark 25 Release. NAG Copyright 2014.

!     .. Use Statements ..
Use nag_library, Only: g13fgf, g13fhf, nag_wp
!     .. Implicit None Statement ..
Implicit None
!     .. Parameters ..
Integer, Parameter :: nin = 5, nout = 6
!     .. Local Scalars ..
Real (Kind=nag_wp) :: hp, lgf, tol
Integer :: i, ifail, ip, iq, l, ldcovr, ldx, &
           lwork, maxit, mn, npar, nreg, nt, num
Logical :: copts, tdist
Character (1) :: dist
!     .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: covr(:,:), et(:), fht(:), ht(:),
                                    sc(:), se(:), theta(:), work(:), &
                                    x(:,:), yt(:)
!     .. Executable Statements ..
Write (nout,*) 'G13FGF Example Program Results'
Write (nout,*)

!     Skip heading in data file
Read (nin,*)

!     Read in the problem size
Read (nin,*) num, mn, nreg

ldx = num
Allocate (yt(num),x(ldx,nreg))

!     Read in the series
Read (nin,*) yt(1:num)

!     Read in the exogenous variables
If (nreg>0) Then
    Read (nin,*)(x(i,1:nreg),i=1,num)
End If

!     Read in details of the model to fit
Read (nin,*) dist, ip, iq

!     Read in control parameters
Read (nin,*) copts, maxit, tol
```

```

!      Calculate NPAR
npar = 1 + 2*iq + ip + mn + nreg
If (dist=='T' .Or. dist=='t') Then
    npar = npar + 1
    tdist = .True.
Else
    tdist = .False.
End If

ldcovr = npar
lwork = (nreg+3)*num + npar + 403
Allocate (theta(npar),se(npar),sc(npar),covr(ldcovr,npar),et(num), &
ht(num),work(lwork))

!      Read in initial values
!      alpha_0
Read (nin,*) theta(1)
l = 2
!      alpha_i and psi_i
If (iq>0) Then
    Read (nin,*) theta(1:(l+iq-1))
    l = l + iq
    Read (nin,*) theta(1:(l+iq-1))
    l = l + iq
End If
!      beta_i
If (ip>0) Then
    Read (nin,*) theta(1:(l+ip-1))
    l = l + ip
End If
!      degrees of freedom
If (tdist) Then
    Read (nin,*) theta(1)
    l = l + 1
End If
!      mean
If (mn==1) Then
    Read (nin,*) theta(1)
    l = l + 1
End If
!      Regression parameters and pre-observed conditional variance
If (.Not. copts) Then
    Read (nin,*) theta(1:(l+nreg-1))
    Read (nin,*) hp
End If

!      Fit the GARCH model
ifail = -1
Call g13fgf(dist,yt,x,ldx,num,ip,iq,nreg,mn,npar,theta,se,sc,covr, &
ldcovr,hp,et,ht,lgf,copts,maxit,tol,work,lwork,ifail)
If (ifail/=0) Then
    If (ifail/=5 .And. ifail/=6) Then
        Go To 100
    End If
End If

!      Read in forecast horizon
Read (nin,*) nt

Allocate (fht(nt))

!      Calculate the volatility forecast
ifail = 0
Call g13fhf(num,nt,ip,iq,theta,fht,ht,et,ifail)

!      Output the results
Write (nout,*) '                               Parameter          Standard'
Write (nout,*) '                               estimates         errors'
!      Output the coefficient alpha_0
Write (nout,99999) 'Alpha', 0, theta(1), se(1)

```

```

l = 2
! Output the coefficients alpha_i and psi_i
If (iq>0) Then
    Write (nout,99999)('Alpha',i-1,theta(i),se(i),i=1,l+iq-1)
    l = l + iq
    Write (nout,99999)(' Psi',i-l+1,theta(i),se(i),i=1,l+iq-1)
    l = l + iq
End If
Write (nout,*)
! Output the coefficients beta_j
If (ip>0) Then
    Write (nout,99999)(' Beta',i-l+1,theta(i),se(i),i=1,l+ip-1)
    l = l + ip
    Write (nout,*)
End If
! Output the estimated degrees of freedom, df
If (dist=='T') Then
    Write (nout,99998) ' DF', theta(l), se(l)
    l = l + 1
    Write (nout,*)
End If
! Output the estimated mean term, b_0
If (mn==1) Then
    Write (nout,99999) ' B', 0, theta(l), se(l)
    l = l + 1
End If
! Output the estimated linear regression coefficients, b_i
If (nreg>0) Then
    Write (nout,99999)(' B',i-l+1,theta(i),se(i),i=1,l+nreg-1)
End If

! Display the volatility forecast
Write (nout,*)
Write (nout,99997) 'Volatility forecast = ', fht(nt)
Write (nout,*)

100 Continue
99999 Format (1X,A,I0,1X,2F16.2)
99998 Format (1X,A,1X,2F16.2)
99997 Format (1X,A,F12.2)
End Program g13fgfe

```

10.2 Program Data

```

G13FGF Example Program Data
100 1 2 :: NUM,MN,NREG
7.53 6.64 7.39 7.15 6.42
6.32 6.98 7.09 6.63 6.93
7.01 5.30 7.86 6.73 7.39
5.61 7.02 6.04 7.46 4.33
6.02 6.37 3.93 7.24 8.58
5.70 9.13 7.99 7.79 6.13
8.78 6.52 6.79 7.77 7.31
7.58 8.78 7.39 8.00 7.07
7.65 9.15 8.32 7.32 7.58
9.78 8.17 9.26 7.79 7.03
7.45 7.09 8.06 7.06 9.91
7.01 8.32 6.41 8.59 8.55
7.77 8.04 9.54 8.28 7.97
8.42 8.30 7.98 7.60 8.77
7.54 7.40 9.26 7.30 9.33
9.54 8.08 6.93 4.27 2.65
5.03 0.91 12.63 10.87 9.26
8.30 6.85 7.48 9.67 9.54
7.33 8.84 7.75 8.12 7.29
8.58 7.80 3.07 9.33 16.91 :: End of Y
2.40 0.12 2.40 0.12
2.40 0.13 2.40 0.14
2.40 0.14 2.40 0.15
2.40 0.16 2.40 0.16

```

```

2.40  0.17      2.41  0.18
2.41  0.19      2.41  0.19
2.41  0.20      2.41  0.21
2.41  0.21      2.41  0.22
2.41  0.23      2.41  0.23
2.41  0.24      2.42  0.25
2.42  0.25      2.42  0.26
2.42  0.26      2.42  0.27
2.42  0.28      2.42  0.28
2.42  0.29      2.42  0.30
2.42  0.30      2.43  0.31
2.43  0.32      2.43  0.32
2.43  0.33      2.43  0.33
2.43  0.34      2.43  0.35
2.43  0.35      2.43  0.36
2.43  0.37      2.44  0.37
2.44  0.38      2.44  0.38
2.44  0.39      2.44  0.39
2.44  0.40      2.44  0.41
2.44  0.41      2.44  0.42
2.44  0.42      2.45  0.43
2.45  0.43      2.45  0.44
2.45  0.45      2.45  0.45
2.45  0.46      2.45  0.46
2.45  0.47      2.45  0.47
2.45  0.48      2.46  0.48
2.46  0.49      2.46  0.49
2.46  0.50      2.46  0.50
2.46  0.51      2.46  0.51
2.46  0.52      2.46  0.52
2.46  0.53      2.47  0.53
2.47  0.54      2.47  0.54
2.47  0.54      2.47  0.55
2.47  0.55      2.47  0.56
2.47  0.56      2.47  0.57
2.47  0.57      2.48  0.57
2.48  0.58      2.48  0.58
2.48  0.59      2.48  0.59
2.48  0.59      2.48  0.60
2.48  0.60      2.48  0.61
2.48  0.61      2.49  0.61
2.49  0.62      2.49  0.62
2.49  0.62      2.49  0.63
2.49  0.63      2.49  0.63
2.49  0.64      2.49  0.64
2.49  0.64      2.50  0.64
'T' 1 2
T 200 0.00001
0.05
-0.15 -0.05
0.05 0.15
0.35
3.25
1.50
4

:: End of X
:: DIST,IP,IQ
:: COPTS,MAXIT,TOL
:: ALPHA_0
:: ALPHA_I
:: PSI_I
:: BETA_I
:: DF
:: MEAN
:: NT

```

10.3 Program Results

G13FGF Example Program Results

	Parameter estimates	Standard errors
Alpha0	0.17	0.19
Alpha1	-0.65	0.23
Alpha2	-0.44	0.24
Psi1	-0.06	0.22
Psi2	0.35	0.25
Beta1	0.42	0.17

DF	5.59	3.75
B0	128.75	42.09
B1	-51.74	17.78
B2	13.01	3.40

Volatility forecast = 1.34
